# Maintainer: Sébastien Luttringer
# Contributor: Andrea Zucchelli <zukka77@gmail.com>

pkgname=iptstate
pkgver=2.2.7
pkgrel=1
pkgdesc='top-like interface to netfilter connection-tracking table'
arch=('loong64' 'x86_64')
url='https://www.phildev.net/iptstate/'
license=('custom')
depends=('glibc' 'gcc-libs' 'ncurses' 'libnetfilter_conntrack')
validpgpkeys=('121BDA2D4ACB63616B367A0E58E11BB1E414D9AD') # Phil Dibowitz
source=("https://github.com/jaymzh/iptstate/releases/download/v$pkgver/iptstate-$pkgver.tar.bz2"{,.asc})
sha256sums=('896df0602885456a2631f795d634fc21311e505f8c910348e6312860f2097955'
            'SKIP')

build() {
  cd $pkgname-$pkgver
  make
}

package() {
  cd $pkgname-$pkgver
  make install SBIN="$pkgdir/usr/bin" MAN="$pkgdir/usr/share/man"
  install -D -m644 LICENSE "$pkgdir/usr/share/licenses/$pkgname/LICENSE"
}
# vim:set ts=2 sw=2 et:
